

Bereichsnamen

Autor & Copyright: Dipl.-Ing. Harald Nahrstedt

Version: 2016 / 2019 / 2021 / 365

Erstellungsdatum: 15.10.2011

Überarbeitung: 01.12.2023

Beschreibung:

Eine der großen Stärken von Excel ist die Vergabe von Namen für Zellen und Zellbereiche. Doch Namen sind auch Objekte, deren Attribute und Methoden sich in VBA nutzen lassen. Dieses Kapitel beschreibt den Umgang mit Namen unter VBA.

Anwendungs-Datei: AE-005_Bereichsnamen.xlsm

1 Die Objektliste Names

Bereichsnamen, oder kurz Namen, stehen als Objektlisten des Workbooks zur Verfügung. Die nachfolgende Prozedur zeigt alle vorhandenen Namen eines Excel-Workbooks, falls vorhanden.

Codeliste 1. Prozedur ZeigeAlleNamen in modNames zeigt alle vorhandenen Bereichsnamen

```
Sub ZeigeAlleNamen()  
    Dim objName As Object  
    Dim sText As String  
  
    sText = ""  
    For Each objName In ThisWorkbook.Names  
        If objName Is Nothing Then  
            sText = sText & vbCrLf & objName.Name  
        End If  
    Next  
    MsgBox sText, vbOKOnly, "Vorhandene Namen"  
End Sub
```

2 Namen erstellen und löschen

Die Erstellung eines Namens ist nicht an einen Zellbereich gebunden. Im nachfolgenden Beispiel wird der Name *MehrwertSteuerProzent* erstellt, und ihm wird der Wert 19% zugewiesen.

Codeliste 2. Die Prozedur NameErstellenOhneBezug im Modul modNames

```
Sub NameErstellenOhneBezug()  
    ActiveWorkbook.Names.Add _  
        Name:="MehrwertSteuerProzent", RefersToR1C1:="=19%"  
    ActiveWorkbook.Names("MehrwertSteuerProzent").Comment = _  
        "MehrwertSteuerProzent ist ein Name mit einem" & _  
        " festen Wert (ohne Zellbezug)"  
End Sub
```

Im Namens-Manager befindet sich nach dem Aufruf der Prozedur *NameErstellenOhneBezug* der Name *MehrwertSteuerProzent*. Mit der Schaltfläche *Bearbeiten* sind im Dialogfenster *Name*, *Kommentar* und *Wert* zu sehen (Bild 1).




Bild 1. Erstellter Name MehrwertSteuerProzent mit konstantem Wert von 19% im Bearbeitungsfenster

Im Namens-Manager kann der Name über die Schaltfläche *Löschen* auch wieder gelöscht werden (Bild 2).




Bild 2. Eintrag im Namens-Manager

Mit diesem Namen lässt sich dann in der ganzen Mappe rechnen (Bild 3).




Bild 3. Formel links, Ergebnis rechts

Das Löschen des Namens zeigt die folgende Prozedur.

Codeliste 3. Die Prozedur NameLöschen im Modul modNames

```
Sub NameLöschen()  
    ActiveWorkbook.Names("MehrwertSteuerProzent").Delete  
End Sub
```

Der häufigere Fall aber ist einen Namen mit Zellbezug anzulegen. Die nachfolgende Prozedur ordnet dem Zellbereich B2:D12 in der Tabelle1 den Namen *Daten1* zu.

Codeliste 4. Die Prozedur NameErstellenMitBezug im Modul modNames

```
Sub NameErstellenMitBezug1()  
    ActiveWorkbook.Names.Add Name:="Daten1", _  
        RefersToR1C1:="=Tabelle1!R2C2:R12C4"  
End Sub
```

Eine andere Möglichkeit ist die nachfolgende Form.

Codeliste 5. Die Prozedur NameErstellenMitBezug2 im Modul modNames

```
Sub NameErstellenMitBezug2()  
    Sheets("Tabelle1").Select  
    Range("E3:H14").Name = "Daten2"  
End Sub
```

3 Bereichs- und Namensobjekte

Auch einem Zellbereichsobjekt lässt sich ein Namensobjekt zuordnen.

Codeliste 6. Die Prozedur NameErstellenFürBereichsObjekt im Modul modNames

```
Sub NameErstellenFürBereichsObjekt()  
    Dim Bereichsobjekt As Object  
  
    Set Bereichsobjekt = _  
        ActiveWorkbook.Sheets("Tabelle1").Range("B10:D15")  
    ActiveWorkbook.Names.Add Name:="Daten3", _  
        RefersTo:= "=" & Bereichsobjekt.Address(External:=True)  
    Set Bereichsobjekt = Nothing  
End Sub
```

Umgekehrt lässt sich ein Namensobjekt einem Zellbereichsobjekt zuordnen.

Codeliste 7. Die Prozedur NameEinemObjektZuordnen im Modul modNames

```
Sub NamenEinemObjektZuordnen()  
    Dim Bereichsobjekt As Object  
  
    Set Bereichsobjekt = _  
        Evaluate(ActiveWorkbook.Names("Daten3").Value)  
    Set Bereichsobjekt = Nothing  
End Sub
```

4 Namen darstellen

Der Bereich, dem ein Name zugeordnet ist, lässt sich aktivieren, so dass dieser auf dem Tabellenblatt gekennzeichnet wird.

Codeliste 8. Die Prozedur NamensbereichAktivieren im Modul modNames

```
Sub NamensbereichAktivieren()  
    Application.Goto Reference:= _  
        ActiveWorkbook.Names("Daten3").Name  
End Sub
```

Der Bereich, dem ein Name zugeordnet ist, lässt sich in der linken oberen Ecke Ihres Tabellenblattes positionieren.

Codeliste 9. Die Prozedur NamensbereichPositionieren im Modul modNames

```
Sub NamensbereichPositionieren()  
    Application.Goto Reference:= _  
        ActiveWorkbook.Names("Daten3").Name, Scroll:=True  
End Sub
```

5 Namenbereiche mit Werten füllen

Wie alle Zellen eines Namensbereichs gefüllt werden, zeigt die nachfolgende Prozedur. Darin werden alle Zellen des Bereichs mit dem Wert 123 gefüllt.

Codeliste 10. Die Prozedur NamensbereichAusfüllen1 im Modul modNames

```
Sub NamensbereichAusfüllen1()  
    Application.Goto Reference:=ActiveWorkbook. _  
        Names("Daten3").Name  
    Selection.Value = 123  
End Sub
```

Einen Namensbereich ohne Aktivierung auszufüllen, schafft die nächste Prozedur. Hier wird der Text xyz in alle Zellen geschrieben.

Codeliste 11. Die Prozedur NamensbereichAusfüllen2 im Modul modNames

```
Sub NamensbereichAusfüllen2()  
    Evaluate(ActiveWorkbook.Names("Daten3").Value).Value = "xyz"  
End Sub
```

Die nachfolgende Prozedur schreibt in alle Zellen des Namensbereichs Zufallszahlen.

Codeliste 12. Die Prozedur NamensbereichAusfüllen3 im Modul modNames

```
Sub NamensbereichAusfüllen3()  
    Dim objZelle As Object  
    Dim dZufall As Double  
  
    Randomize  
    dZufall = Time  
    For Each objZelle In _  
        ActiveWorkbook.Names("Daten3").RefersToRange.Cells  
        dZufall = Rnd(dZufall)  
        objZelle = dZufall  
    Next  
End Sub
```

Aber auch einzelne Zellen eines Namensbereichs lassen sich beschreiben. In der nachfolgenden Prozedur wird in die erste Zeile und zweite Spalte des Namensbereichs der Wert 33 geschrieben.

Codiliste 13. Die Prozedur NamensbereichAusfüllen4 im Modul modNames

```
Sub NamenbereichAusfüllen4()  
    ActiveWorkbook.Names("Daten3").RefersToRange.Cells(1, 2) = 33  
End Sub
```